

SelfBank
by Singular Bank

PROSCREENER

LA HERRAMIENTA QUE TE AYUDARÁ
A RASTREAR EL MERCADO



ÍNDICE

PRESENTACIÓN DE PROSCREENER	03
CAPÍTULO I : NOCIONES FUNDAMENTALES	04
Acceder a ProScreeener	04
Las ventanas TopVariación	09
CAPÍTULO II: PROGRAMAR CON PROSCREENER	10
Instrucciones de búsqueda y filtrado	10
Anticipar el volumen	13
Búsqueda en períodos múltiples	13
Búsqueda de múltiples valores	15
CAPÍTULO III: APLICACIONES PRÁCTICAS	18
Ejemplos simples	18
Ejemplos más sofisticados	20

PRESENTACIÓN DE PROSCREENER

ProScreener es una sofisticada herramienta de rastreo que le permitirá rastrear mercados enteros:

- Buscando valores que cumplan una o varias condiciones definidas por usted mismo(a).
- En una o múltiples unidades de tiempo (p.ej.: 1 minuto y 1 hora).
- En tiempo real o al cierre de cada vela, con una precisión de tick por tick.

ProScreener integra el lenguaje de programación ProBuilder con extensiones que se aplican exclusivamente a las condiciones de filtrado. Los rastreos en tiempo real o al cierre de la vela pueden aplicarse en las siguientes unidades de tiempo:

- 1 minuto.
- 2 minutos.
- 3 minutos.
- 5 minutos.
- 10 minutos.
- 15 minutos.
- 30 minutos.
- 1 hora.
- 2 horas.
- 3 horas.
- 4 horas.
- Diario.
- Semanal.

Los resultados de un escaneo ProScreener se muestran en una lista de resultados con los 30 mejores valores según el criterio de ordenación definido.

Este documento debe considerarse como una extensión del manual ProBuilder aunque también puede leerse de manera independiente. Mediante un enfoque teórico y ejemplos concretos, podrá aprender las bases de la programación de códigos que le permitirán definir filtros aplicables sobre cualquier mercado incluido en la cobertura de Trader Pro8.02.


En el final de este manual, encontrará un Glosario que le proporciona un listado completo de todos los comandos que pueden utilizarse en este módulo.

¡Le deseamos una buena lectura y mucho éxito en sus inversiones!

CAPÍTULO I : NOCIONES FUNDAMENTALES

ACCEDER A PROSCREENER

Puede acceder al módulo ProScreener pulsando en 'Mostrar' -> 'ProScreener', o con las teclas Ctrl+E:



Esto hará aparecer la ventana ProScreener, donde puede seleccionar un código ya existente o programar un código nuevo. Para ello, acceda a la ventana de programación pulsando 'Gestión ProScreeners' -> 'Nuevo ProScreener'. Encontrará dos opciones:

- La 'Creación asistida', que le permite definir condiciones de búsqueda sin necesidad de teclear código. Para más información, consulte el vídeo 'Programa su ProScreener sin necesidad de teclear código').
- La 'Creación programando', que le permite definir condiciones más sofisticadas y sus parámetros de aplicación mediante código.

PROSCREENER

La ventana 'Creación programando' contiene 5 secciones:

- Zona de programación del ProScreener.
- Selección del mercado en el que se efectuará el rastreo.
- Período considerado.
- Criterio de clasificación de resultados.
- Zona para compartir el código con la comunidad de usuarios de Trader Pro8.02.

The screenshot shows the ProScreener application window with the 'Creación programando' tab selected. The interface is divided into five main sections:

- 1 - Programación del ProScreener:** Contains two buttons: 'Insertar función' and 'Ayuda ProScreener'. Below them is a large empty text area for entering functions.
- 2 - Selección de lista:** Features radio buttons for 'Listas predefinidas' (selected) and 'Listas personalizadas'. Below is a dropdown menu showing 'IBEX 35'.
- 3 - Período:** A dropdown menu set to '1 hora'.
- 4 - Clasificación de los 30 mejores resultados:** Radio buttons for 'Mostrar valores con mayor cumplimiento de los criterios' (selected) and 'Mostrar valores con menor cumplimiento de los criterios'.
- Ayuda ProScreener:** A text box providing advice on how to publish a ProScreener, including details on usage, market conditions, and expectations.
- 5 - Mostrar ProScreener:** A checkbox for 'Mantener ventana abierta tras validación' and two buttons: 'Validar programa' and 'Cancelar'.
- Compartir ProScreener:** Three checkboxes: 'Publicar en la sección 'Sistemas' de la Web ProRealTime', 'Ocultar código (permitiendo uso si no se restringe)', and 'Restringir uso mediante contraseña:' followed by a password input field.

En la **primera** sección podrá:

PROSCREENER

- Programar directamente un ProScreener en la zona de texto
- Utilizar la función de ayuda 'Insertar función', que permite hallar en una nueva ventana la librería de funciones clasificadas en 7 categorías.

Empezaremos utilizando la opción 'Insertar función' pulsando en el botón correspondiente.

Tomemos como ejemplo la instrucción '**SCREENER**', que permite buscar condiciones. Esta instrucción se encuentra en la categoría 'Comandos ProScreener'.

Pulse en la palabra '**SCREENER**' -> 'Añadir': el comando aparecerá en la zona de programación.

The screenshot shows the ProScreener application window. At the top, the title bar reads 'ProScreener'. Below it, there is a text field for 'Nombre ProScreener' containing 'Busqueda'. Two tabs are visible: 'Creación asistida' and 'Creación programando', with the latter being active. The main area is divided into several sections:

- 1 - Programación del ProScreener:** Contains two buttons: 'Insertar función' and 'Ayuda ProScreener'. Below them is a large text area containing the word 'SCREENER'.
- 2 - Selección de lista:** Features radio buttons for 'Listas predefinidas' (selected) and 'Listas personalizadas'. Below is a dropdown menu showing 'IBEX 35'.
- 3 - Período:** A dropdown menu showing '1 hora'.
- 4 - Clasificación de los 30 mejores resultados:** Radio buttons for 'Mostrar valores con mayor cumplimiento de los criterios' and 'Mostrar valores con menor cumplimiento de los criterios'.
- Ayuda ProScreener:** A text box containing instructions: 'De cara a una eventual publicación de su ProScreener, le aconsejamos incluya la siguiente información: - uso (entrar a largo, vender a corto, ...); - situaciones en las que se puede aplicar (volatilidad alta o baja, ...); - mercados o listas en los que funciona mejor; - expectativas de resultados (en capital inicial, curva de liquidez, ...); - cualquier información útil suplementaria'.
- 5 - Mostrar ProScreener:** A checkbox for 'Mantener ventana abierta tras validación' and two buttons: 'Validar programa' and 'Cancelar'.
- Compartir ProScreener:** Three checkboxes: 'Publicar en la sección 'Sistemas' de la Web ProRealTime', 'Ocultar código (permitiendo uso si no se restringe)', and 'Restringir uso mediante contraseña:' followed by a text input field.

PROSCREENER

El comando '**SCREENER**' le permite definir la(s) condición(es) de búsqueda. La última línea de cualquier ProScreener deberá comenzar por el comando '**SCREENER**'.

Supongamos que deseamos buscar todos los valores del Ibex español en los que la apertura de la vela (período) actual sea mayor que el cierre de la vela precedente. El código correspondiente sería:

```
c1 = (Open > Close [1])  
SCREENER [c1]
```

Una vez definido el código, vamos a elegir la segunda sección, una lista predefinida o una personalizada en la que efectuar la búsqueda.

Nota: La búsqueda se realiza exclusivamente en valores de un único mercado. Para aplicar un ProScreener a una lista personalizada, deberá asegurarse de que todos los valores que la componen pertenezcan al mismo mercado.

La **tercera** sección le permite definir el período utilizado durante el rastreo del ProScreener. Dado que la cotización puede verse en velas de períodos distintos, una configuración que se dé en velas de una unidad de tiempo determinada podría no darse en otra unidad de tiempo distinta.

La **cuarta** sección controla la clasificación de los resultados en función del grado de intensidad del cumplimiento de los criterios definidos en el código (ver la sección: "Instrucciones de búsqueda y filtrado"). Si las condiciones son poco restrictivas, los resultados pueden ser muy amplios (llegando a varios millares en el caso de mercados con muchos valores cotizados, como el NASDAQ. Por ello, es conveniente filtrar los resultados según uno de estos criterios:

- Los 30 valores que cumplen los criterios con mayor intensidad.
- Los 30 valores que cumplen los criterios con menor intensidad.

Ejemplo: supongamos que su ProScreener rastrea todo el NASDAQ buscando en velas diarias todos los valores que tengan un volumen diario superior a 20000. Cabe esperar que el número de resultados de tal búsqueda será superior a la treintena. Así, parametrar el criterio de filtraje por ejemplo en el % de variación de los valores le permitirá visualizar los resultados con las variaciones más amplias o las menos amplias.

La **quinta** sección le permite compartir el ProScreener creado con nuestra comunidad de usuarios.

PROSCREENER

Tras definir estos 5 parámetros, pulse 'Validar programa' para lanzar la búsqueda. Si ésta proporciona resultados, se mostrarán en una lista similar a ésta:

ProScreener

Busqueda Gestión ProScreeners

Num total de resultados (se muestra un máximo de 30): 35

↑ Nombre	Ultimo	% Var	Volumen	Apertura
UNION FENOSA	16,93	-0,41%	255.761	16,91
TELEFONICA	15,30	-0,97%	3.152k	15,24
TELECINCO	6,53	-0,91%	81.361	6,60
SANTANDER	8,46	-1,86%	6.385k	8,55
SACYR VALLEH	8,06	+0,75%	48.906	8,00
REPSOL YPF	15,76	-2,29%	847.361	15,72
RED ELE.CORP	33,20	-2,67%	104.845	33,17
OBR.H.LAIN	10,90	-0,91%	24.380	11,00
MAPFRE	2,52	-3,08%	1.149k	2,47
INDRA A	15,62	-1,45%	34.818	15,77
INDITEX	27,64	-1,74%	74.300	27,60
IBERIA	1,82	-1,09%	367.810	1,84
IBERDROLA	6,05	-2,26%	2.802k	6,14
IBE.RENOVABL	2,60	-1,14%	1.384k	2,66
GRIFOLS	14,97	-0,20%	160.940	15,00
GAS NATURAL	25,67	-1,80%	96.298	25,97
GAMESA	14,62	-4,32%	338.002	15,25
G.FERROVIAL	25,39	-1,55%	46.254	25,90
FCC	31,61	-1,83%	47.363	32,00
ENDESA	26,60	-0,34%	42.052	27,06
ENAGAS	14,93	-0,86%	56.837	15,10
BBVA	9,87	-0,80%	2.391k	9,89
BANKINTER	8,97	-1,43%	128.446	9,00
BANESTO	10,04	-1,67%	17.936	10,01
B.POPULAR	7,39	-1,34%	309.452	7,45
B. SABADELL	5,48	-0,90%	398.551	5,50
ACS CONST.	29,90	-1,52%	34.336	30,21
ACERINOX	10,74	-2,36%	49.887	10,93
ACCIONA	80,15	-1,90%	25.504	81,90
ABERTIS INFR	14,37	-0,55%	100.064	14,35

LAS VENTANAS TOPVARIACIÓN

Además del módulo ProScreener, Trader Pro 8.02 le ofrece versiones predefinidas con un abanico de búsquedas en las ventanas TopVariación. Estas ventanas rastrean los mercados con criterios predefinidos (estos criterios son completamente personalizables en el módulo ProScreener). Ambas ventanas son idénticas y le permitirán lanzar búsquedas simultáneas en el mismo o en distintos mercados (ejemplo: variación al alza en una ventana y a la baja en la otra).

Los criterios de búsqueda propuestos en estas ventanas se agrupan en 4 categorías, capaces de detectar:

- Variaciones y gaps.
- Las mayores variaciones en la preapertura del mercado.
- Las principales configuraciones de velas japonesas.
- Los valores de Spread y Volumen en el libro de órdenes de compraventa.

Top Variación 1

Variaciones Preapertura Velas (candlesticks) Profundidad de mercado

Mercado: Mercado Continuo Variación: Subidas Desde: Ayer

Nº de resultados: 30

Nombre	↑ % Var Ayer	Volumen	Precio ref
FERSA	+12,97%	65.914	2,39
SACYR VALLEH	+9,00%	69.321	8
PRIM	+4,80%	400	6,25
SOLARIA	+3,73%	361.643	2,68
INDO	+3,70%	23.129	1,35
PARQUESOL	+3,58%	435	9,5
CIE AUTOMOT.	+3,54%	1.589	3,67
SERV.POINT S	+3,16%	97.498	0,95
EUROPAC	+2,90%	4.121	3,79
INBESOS	+2,33%	38	3,87
BAYER	+2,27%	25	44
ITINERE	+2,00%	800	4
TUBACEX	+1,94%	109.893	3,1
AG.BARCELONA	+1,82%	4.457	16,45
MONTEBALITO	+1,62%	500	4,94
GRAL.ALQ.MAQ	+1,40%	3.956	6,44

Para saber más sobre el uso de TopVariación, consulte la rúbrica 'Top Variación' en la sección de Ayuda de nuestra Web, o el vídeo 'Visualice las mayores variaciones en la Preapertura, precio o volumen con las ventanas de Top Variación'.

CAPÍTULO II: PROGRAMAR CON PROSCREENER

En este capítulo le mostramos los 5 comandos ProBuilder específicos al módulo ProScreener. Los encontrará también en la biblioteca de funciones, en la sección 'Comandos ProScreener'. Vamos a revisar los siguientes:

- Instrucciones de búsqueda y filtrado.
- Anticipar el volumen.
- Búsqueda en períodos múltiples.
- Búsqueda de múltiples valores.

INSTRUCCIONES DE BÚSQUEDA Y FILTRADO

El comando '**SCREENER**' lanza la búsqueda. Es el equivalente a '**RETURN**' en el lenguaje ProBuilder; al igual que éste, puede acompañarse de muchas funciones complementarias que vamos a estudiar a continuación.

'**SCREENER**' se utiliza de la siguiente manera:

SCREENER [Condición]

Tomemos un ejemplo:

Vamos a buscar valores de un mercado en los que el precio al cierre sea estrictamente menor a la banda Bollinger inferior. Las bandas de Bollinger se calculan sobre 10 periodos y se aplican al precio al cierre.

```
c1 = (Close < BollingerDown [10] (Close))  
SCREENER [c1]
```

Podemos ampliar la búsqueda a condiciones múltiples, y establecer que éstas se cumplan simultáneamente o alternativamente. En tal caso, la sintaxis de ProScreener sería:

```
SCREENER [Condición 1 AND Condición 2]
```

en el caso de que todas las condiciones deban cumplirse simultáneamente, o bien

```
SCREENER [Condición 1 OR Condición 2]
```

si queremos establecer que al menos una de las condiciones deba cumplirse.

PROSCREENER

Veamos un ejemplo:

Busquemos valores en los que el precio sea mayor que el canal de Bollinger y que tengan una tendencia al alza.

REM Cierre por encima de la Boll sup

Condition 1 = `(Close > Bollingerup [20](Close))`

REM Cierre > a laApertura

Condition 2 = `Close > Open`

REM MM7 > MM23

Condition 3 = `(Average [7](Close) > Average [23](Close))`

SCREENER [Condition 1 AND Condition 2 AND Condition 3]

También podemos definir una constante o un indicador (predefinido o personalizado) que ejercerá simultáneamente de criterio de filtrado y de clasificación de resultados. Así,

- Si la búsqueda proporciona más de 30 resultados, el indicador filtra los valores que serán visualizados en la lista. En este caso, puede utilizar la cuarta sección de la ventana de programación del ProScreener para definir el orden de aparición de estos resultados (ver la descripción en la p.5).
- Si el número de resultados es igual o menor a 30, el indicador los ordena según la base de este nuevo indicador.

La sintaxis para insertar una constante es:

SCREENER [c1] (TypicalPrice)

En el caso de un indicador predefinido, deberemos comenzar por almacenarlo en una variable (que en este caso, llamaremos 'Criterio'):

Criterio = `RSI [14](Close) SCREENER [c1](Criterio)`

Si utilizamos un indicador personalizado, utilizaremos la instrucción 'CALL' (cuya definición se encuentra en el manual ProBuilder).

MiRSI = `CALL "RSI" [14] SCREENER [c1](MiRSI)`

PROSCREENER

Ejemplo: Busquemos todos los valores de un mercado cuyo volumen de negociación sea mayor que 50000. Crearemos un indicador que nos permita evaluar si el RSI se encuentra en situación de sobrecompra o sobreventa. Mostraremos los resultados del ProScreener ordenados a partir de dicho indicador.

```
c1 = Volume > 50000
IF RSI [20](Close) > 70 THEN
    Criterio = 1
```

```
ELSIF RSI [20](Close) < 30 THEN
    Criterio = -1
ENDIF
```

SCREENER [c1] (Criterio AS "sobrecompra")

Nombre	% Var	Volumen	↑ sobrecompra
BBVA	-0,90%	3.533k	1
GAMESA	-4,38%	493.638	1
GAS NATURAL	-2,95%	116.610	1
IBERDROLA	-1,45%	3.891k	1
IBERIA	-1,63%	550.015	1
INDITEX	-1,81%	102.841	1
B.POPULAR	-1,34%	391.132	1
RED ELE.CORP	-1,20%	107.031	1
REPSOL YPF	-2,91%	1.029k	1
TELEFONICA	-0,65%	3.765k	1
ZELTIA	-1,22%	161.982	1
ENAGAS	-1,06%	82.809	1
ABERTIS INFR	-0,14%	128.841	1
SACYR VALLEH	+5,25%	101.077	1
MAPFRE	-1,54%	1.563k	1
PRISA	-4,00%	128.056	1
ERCROS	+0,00%	1.669k	1
B. SABADELL	-0,36%	561.941	1
LA SEDA	-2,08%	1.279k	-1
SOLARIA	+1,12%	399.278	-1
SNIACE	-0,90%	70.540	-1
SERV.POINT S	+3,16%	105.110	-1
GRIFOLS	-0,33%	191.477	-1

También podemos omitir el uso de paréntesis introduciendo el comando '**SORTBY**':

SCREENER [c1] SORT BY TypicalPrice

PROSCREENER

Si queremos atribuir un nombre personalizado esta columna, deberemos utilizar la función 'AS':

```
SCREENER [c1](TypicalPrice AS "PrecioTipico")
```

ANTICIPAR EL VOLUMEN

El comando 'EstimatedVolume' proporciona una estimación lineal del volumen de la barra actual. Para ser exactos, se calcula el volumen estimado a partir de la siguiente fórmula:

$$\text{Volumen_estimado} = \text{Volumen} * \text{Coeficiente_multiplicador}$$

o

$$\text{Coeficiente_multiplicador} = \text{duración de la vista} / \text{tiempo transcurrido desde la creación de la vela}$$

Este comando es especialmente interesante para comparar el volumen estimado con el volumen real.

Veamos un ejemplo: En vista 10 minutos, si suponemos que la última vela se está formando desde hace 1 minuto, el volumen estimado será 10 veces el volumen actual.

Calculemos la relación entre el volumen anticipado del día respecto al volumen de la víspera (para obtener resultados representativos, es aconsejable utilizar un período intradiario):

REM Evaluacion del volumen del día

Vol0 = [EstimatedVolume](#)

REM Hallar el volumen de la víspera

Vol1 = [Volume \[1\]](#)

REM Criterio de clasificación: relación entre el volumen anticipado del día respecto al volumen de la víspera

[SCREENER \(Vol0 / Vol1 AS "Volumen"\)](#)

BÚSQUEDA EN PERÍODOS MÚLTIPLES

Puede definir una búsqueda que verifique los criterios en múltiples unidades de tiempo. Esto permite comprobar que una condición se satisfaga, por ejemplo a corto o medio plazo. El comando adaptado se denomina 'TIMEFRAME', y su sintaxis es la siguiente:

TIMEFRAME (unidad de tiempo del período)

PROSCREENER

Las unidades de tiempo disponibles se traducen en código por:

Interpretación	Código
1 minuto	1 minute
2 minutos	2 minutes
3 minutos	3 minutes
5 minutos	5 minutes
10 minutos	10 minutes
15 minutos	15 minutes
30 minutos	30 minutes
1 hora	1 hour
2 horas	2 hours
3 horas	3 hours
4 horas	4 hours
Diario	Daily
Semanal	Weekly

Las instrucciones que siguen a la línea **'TIMEFRAME'** se buscarán únicamente en el período indicado. Es posible crear varias líneas **'TIMEFRAME'** en un mismo código, para efectuar una búsqueda en períodos múltiples.

Tomemos un ejemplo: Busquemos todos los valores del Ibex 35 que cumplan los siguientes criterios:

- En vista semanal, el indicador Williams %R de 14 períodos debe tener un valor entre 0 y-20.
- En vista de 30 minutos, la media móvil exponencial de 20 períodos cruza al alza la de 12 períodos.

Podríamos detectar los valores situados en zona de sobrecompra en períodos semanales y que presenten a corto plazo una tendencia bajista.

TIMEFRAME (weekly)

Condition1 = Williams [14](Close) < 0 AND Williams [14](Close) > -20

TIMEFRAME (30minutes)

Condition 2 = $\text{ExponentialAverage}[20](\text{Close})$ **CROSSES OVER**
 $\text{ExponentialAverage}[12](\text{Close})$
SCREENER [Condicion 1 AND Condicion 2]

BÚSQUEDA DE MÚLTIPLES VALORES

Hemos constatado que con ProScreener se debe especificar un mercado o una lista de valores sobre el que efectuar la búsqueda.

La instrucción '**EQUITYFRAME**' permite relacionar el criterio con un valor específico perteneciente al mismo mercado en el que se realiza la búsqueda.

Se utiliza de esta manera:

`EQUITYFRAME ("nombre del mercado" ,"ticker")`

El comando compara los resultados con un valor concreto y permite la creación de un nuevo indicador que puede utilizarse como condición de ordenación. El uso optimizado de este comando requiere que se respeten estas condiciones:

- Debe emplearse exclusivamente en mercados en los que se disponga de acceso a datos en tiempo real
- No puede aplicarse a los índices de los mercados
- El valor elegido debe formar parte de la misma lista de búsqueda (ver sección 2 de la ventana Proscreeener)

Así, para visualizar una treintena de valores del mercado Ibex 35 seleccionados aleatoriamente y relacionados con el precio de cierre actual del valor TELEFONICA (ticker 'TEF'), el código será:

`EQUITYFRAME("Acciones Espana" ,"TEF")`
`MiCierre =Close`
SCREENER(MiCierre)

Los mercados siguen la sintaxis descrita a continuación:

Código EQUITYFRAME	Nombre del mercado
AccionesItalia	AccionesItalia
Acciones & ETFsAMEX	AccionesAmex
AccionesAustralia	AccionesAustralia
AccionesHolanda	EuronextAmsterdam
AccionesBélgica	EuronextBruselas
AccionesPortugal	EuronextLisboa
AccionesFrancia	EuronextParis
Forex (47pares)	Divisas /monedas
LSE	London StockEchange
AccionesEspana	Acciones España
AccionesNASDAQ	AccionesNasdaq
AccionesNYSE	AccionesNyse
Acciones SuizaVirtX	AccionesSuiza
Acciones SuizaSWX	AccionesSuiza
AccionesAlemania	AccionesAlemania

Al igual que en el caso del comando **'TIMEFRAME'**, el código que sigue a **'EQUITYFRAME'** se aplica al valor especificado por este último. Para volver a los datos del valor seleccionado en la interfaz de ProScreener, podemos escribir:

EQUITYFRAME (default)

PROSCREENER

Las dos funciones descritas son complementarias, tal y como se ilustra en el siguiente ejemplo:

Si deseamos visualizar el indicador que muestra la diferencia de precio al cierre de la vela actual del valor TELEFONICA (ticker 'TEF') respecto al precio de cada uno de treinta valores del IBEX 35 seleccionados al azar, el código sería el siguiente:

REM Comencemos por recuperar las informaciones a extraer del valor del IBEX 35

EQUITYFRAME ("Acciones Espana" ,"TEF")

MiCierre = Close

REM Volvemos a los valores del mercado seleccionados en la lista desplegable "Selección de la lista"

EQUITYFRAME (default)

REM Recuperamos las informaciones deseadas

ValCierre = MiCierre - Close

REM Lanzamos la visualización del resultado

SCREENER (ValCierre AS "MiIndicador")

Ejemplo: Vamos a representar visualmente la correlación entre un valor y su mercado, o entre un índice y su mercado. Calculamos el ratio de dos valores seleccionados ('EQUITYFRAME') y acto seguido, la diferencia de este ratio respecto a la víspera.

TIMEFRAME(Daily)

ValCierre = Close

EQUITYFRAME("NASDAQ" ,"AMZN")

IndCierre = Close

EQUITYFRAME(default)

Ratio = (ValCierre / IndCierre) * 100

FuerzaRelativa = (Ratio - Ratio[1]) * 100

SCREENER (FuerzaRelativa AS "Fuerza Relativa")

CAPÍTULO III: APLICACIONES PRÁCTICAS

EJEMPLOS SIMPLES

1. RSI 1 hora: sobreventa

El RSI es un indicador de sobrecompra/sobreventa que puede anticipar un cambio de tendencia. En este ejemplo, vamos a intentar detectar valores que se encuentren en situación de sobreventa.

Definiremos la situación de sobreventa como $RSI < 30$. Cuanto más se acerque a cero el RSI, la situación de sobreventa será más acentuada. Construiremos un ProScreener que nos busque los valores cuyo RSI sea menor que 30.

Ejemplo RSI en sobreventa:

REM Cálculo del RSI de 14 periodos

MiRSI = RSI [14]

REM Filtro: RSI < 30

Filtro = MiRSI < 30

SCREENER [Filtro] (MiRSI AS "RSI")

2. RSI 1 hora: sobrecompra

Busquemos ahora valores en situación de sobrecompra, definida como $RSI > 70$). Cuanto más se acerque al RSI a 100, la situación de sobrecompra será más acentuada. Construiremos un ProScreener que nos busque los valores cuyo RSI se acerque a 100.

Ejemplo RSI en sobrecompra:

REM Cálculo del RSI de 14 periodos

MiRSI = RSI [14]

REM Filtro: RSI > 70

Filtro = MiRSI > 70

SCREENER [Filtro] (MiRSI AS "RSI")

3. Cruce alcista de medias móviles

En este caso buscaremos resaltar una tendencia en la que una media móvil corta cruza una media móvil larga.

Construiremos un ProScreener que busque valores en los que la media móvil de 20 períodos cruce al alza la media móvil de 50 períodos.

Además, calcularemos el momento de la diferencia entre ambas medias móviles para evaluar la fuerza del cruce. Si este número es cercano a cero, el crecimiento será lento y las dos medias móviles deberían ser casi paralelas; la señal será poco significativa. A la inversa, cuanto más elevado sea este número, más 'dinámico' será el cruce.

Un ejemplo típico lo constituye el de una media móvil plana a la que cruza otra media móvil con una fuerte tendencia al alza.

Ejemplo: Cruce al alza de la media móvil a 50 períodos por la media móvil a 20 períodos.

REM Cálculo de la media móvil simple de 20 periodos

MM20 = **Average** [20]

REM Cálculo de la media móvil simple de 50 periodos

MM50 = **Average** [50]

REM Evaluación de la velocidad relativa de la MM corta respecto a la larga

Velocidad = **Momentum** (MM50 - MM20) * 100/Close

REM Selección del valor en el momento del cruce

Filtro = MM20 **Crosses Over** MM50

SCREENER [Filtro] (Velocidad AS "Dinamica")

4. Cruce bajista de medias móviles

Construiremos un ProScreener que busque valores en los que la media móvil de 20 períodos cruce a la baja la media móvil de 50 períodos.

El momento se define de modo análogo al descrito en el ejemplo precedente.

Ejemplo: Cruce a la baja de la media móvil a 50 períodos por la media móvil a 20 períodos

REM Calculo de la media móvil simple de 20 periodos

MM20 = **Average** [20]

REM Calculo de la media móvil simple de 50 periodos

MM50 = **Average** [50]

REM Evaluación de la velocidad relativa de la MM corta respecto a la larga

Velocidad = **Momentum** (MM20 - MM50) * 100/Close

REM Selección del valor en el momento del cruce

Filtro = MM20 **Crosses Under** MM50

SCREENER [Filtro] (Velocidad AS "Dinamica")

EJEMPLOS MÁS SOFISTICADOS

1. RSI y cambio de tendencia alcista

Presentaremos aquí un sistema que permita la detección en tiempo real de los valores más susceptibles de cambiar de tendencia, y creando además un movimiento de gran envergadura.

Tradicionalmente, los analistas se interesan a los indicadores de sobrecalentamiento o a las figuras de cambio de tendencia aislándolas de su contexto (a menudo, por ausencia de herramientas que les permitiesen obrar de otro modo mejor).

Con el módulo ProScreener estas limitaciones técnicas desaparecen, y podemos construir un auténtico sistema de detección basado en el RSI. El punto de inicio es sencillo: buscamos un RSI en sobreventa con un cambio de tendencia al alza.

Introduciremos un filtro que contenga estas dos condiciones. Pueden inscribirse en una sola línea de código:

$$\text{Filtro} = \text{RSI} < 30 \text{ AND Momentum}[1](\text{RSI}) > 0$$

Una vez que tenemos los valores que puedan interesarnos, vamos a ordenarlos para mostrar aquellos que tengan un máximo interés. Para ello, consideraremos aquellos cuya caída haya sido más severa en relación a la volatilidad habitual del valor.

Ejemplo: RSI y cambio de tendencia alcista

```
REM Filtrado de valores con RSI en sobreventa y en cambio de tendencia al alza
// Criterio de filtrado: RSI < 30 y creciente
Filtro = RSI < 30 AND Momentum [1](RSI) > 0
REM Determinar la "fuerza" de la tendencia bajista
REM Hallar el punto máximo en las 20 últimas velas
PuntoMaximo = Highest [20](High)
REM Determinar la caída desde este punto
Caída = PuntoMaximo - Close
REM Determinar la volatilidad habitual del valor (mediana del True Range sobre 3 velas)
Norma = Summation [3](TR) - Highest [3](TR) - Lowest [3](TR)
Velocidad = Caída/Norma
REM Mostrar resultados: bajadas relativas más importantes
SCREENER [Filtro] (Velocidad AS "Fuerzabajista")
```

2. RSI y cambio de tendenciabajista

El punto de inicio de este ProScreener es la búsqueda de un RSI en sobrecompra que experimenta un cambio de tendencia a la baja.

Al igual que en el caso anterior, crearemos un filtro en código ProBuilder:

Filtro = `RSI > 70 AND Momentum[1](RSI) < 0`

Al igual que con el ProScreener previo, recuperaremos los valores en los que la subida sea más marcada en relación con la volatilidad habitual del valor.

Ejemplo: RSI y cambio de tendencia bajista

REM Filtrado de valores en los que el RSI está en sobrecompra y cambio de tendencia a la baja

// Criterio de filtrado: RSI > 70 y decreciente

Filtro = `RSI > 70 AND Momentum [1](RSI) < 0`

REM Determinar la "fuerza" de la tendencia alcista

// Hallar el punto mínimo en las 20 últimas velas previas

PuntoMinimo = `Lowest [20](Low)`

// Determinar la subida desde este punto

Alza = `Close - PuntoMinimo`

// Determinar la volatilidad habitual del valor (mediana del True Range sobre 3 velas)

Norma = `Summation [3](TR) - Highest [3](TR) - Lowest [3](TR)`

Velocidad = `Alza/Norma`

REM Mostrar resultados: subidas relativas más importantes

SCREENER [Filtro](Velocidad AS "Fuerza alcista")

3. Vela envolvente alcista

La vela envolvente alcista es una de las configuraciones de velas japonesas más conocidas, pero que evidentemente ha de ser tomada en cuenta con su contexto.

Empecemos por definir una vela envolvente alcista:

- Vela previa de color negro/rojo.
- Apertura de la vela actual por debajo del cuerpo de la vela previa.
- Cierre de la vela actual por encima del cuerpo de la vela previa.

El filtro que agrupa las tres condiciones será el siguiente:

$$\text{Filtro} = \text{Close [1]} < \text{Open [1]} \text{ AND } \text{Open} < \text{Close [1]} \text{ AND } \text{Close} > \text{Open [1]}$$

En cuanto a la detección de la tendencia bajista, retomaremos el código de los ejemplos anteriores ajustando ligeramente el horizonte temporal. Con el RSI 14 era razonable interesarse a la caída de precios sobre las 20 velas precedentes. En el caso de las velas japonesas, la experiencia indica que basta con estudiar las 8 últimas velas (ya que se trata de configuraciones más reactivas y teóricamente limitadas a un máximo de 7 velas).

Ejemplo: Vela envolvente alcista

```
REM Determinar la "fuerza" de la tendencia bajista
// Buscar el máximo de las 8 velas previas
Maximo = Highest [8](High)
// Determinar la caída desde este punto
Caida = Maximo - Close
// Determinar la volatilidad habitual del valor (mediana del True Range sobre 3 velas)
Norma = Summation [3](TR) - Highest [3](TR) - Lowest [3](TR)
REM Criterio de filtrado: vela envolventealcista
Filtro = Close [1] < Open [1] AND Open < Close [1] AND Close > Open [1]
Velocidad = Caida/Norma
REM Mostrar resultados: caídas relativas mas importantes
SCREENER [Filtro] (Velocidad AS "Fuerza de la caída")
```

4. Vela envolvente bajista

Empecemos por definir una vela envolvente bajista:

- Vela previa de color blanco/verde.
- Apertura de la vela actual por encima del cuerpo de la vela previa.
- Cierre de la vela actual por debajo del cuerpo de la vela previa.

El filtro que agrupa las tres condiciones será el siguiente:

$$\text{Filtro} = \text{Close [1]} > \text{Open [1]} \text{ AND } \text{Open} > \text{Close [1]} \text{ AND } \text{Close} < \text{Open[1]}$$

Para detectar la tendencia alcista, retomaremos el código del RSI con cambio de tendencia a la baja que habíamos creado previamente.

Ejemplo: Vela envolvente bajista

REM Determinar la "fuerza" de la tendencia alcista

// Buscar el mínimo de las 8 velas previas

Minimo = `Lowest [8](Low)`

// Determinar el alza desde este punto

Alza = `Close - Mínimo`

// Determinar la volatilidad habitual del valor (mediana del True Range sobre 3 velas)

Norma = `Summation [3](TR) - Highest [3](TR) - Lowest [3](TR)`

REM Critère de filtrage: avalementbaissier

Filtro = `Close [1] > Open [1] AND Open > Close [1] AND Close < Open [1]`

Velocidad = `Alza/Norma`

REM Mostrar resultados: alzas relativas mas importantes

SCREENER [Filtro](Velocidad AS "Fuerza alcista")

5. Triple pantalla alcista

Este ProScreener se compone de tres condiciones en varias unidades de tiempo:

- **Condición 1:** MACD en vista semanal <0
- **Condición 2:** MACD **creciente** en vista semanal
- **Condición 3:** Estocástico en vista diaria <30

Buscamos los valores más cercanos del nivel de compra STOP recomendado por el sistema (en este caso, se tratará del valor máximo de la víspera).

El ProScreener mostrará los valores situados por debajo de este nivel, así como aquellos que acaben de cruzarlo y que se mantienen en un margen cercano (no superior a un+5%).

Los resultados respetarán las condiciones en las unidades de tiempo indicadas (vista semanal y diaria).

Ejemplo: Triple pantalla alcista

REM Condicion 1: MACD semanal < 0 y creciente

TIMEFRAME(weekly)

MiMACD = MACD [12,26,9](Close)

c1 = MiMACD < 0 AND MiMACD > MiMACD [1]

REM Condicion 2: Estocastico diario < 30

TIMEFRAME(daily)

MiSTO = Stochastic [14,3](Close)

c2 = MiSTO < 30

REM Nivel del stop de compra

MiStop = High [1]

REM Criterio: Posicion de los precios respecto al STOP

Criterio = (Close/MiStop - 1) *100

REM Condicion 3: los precios bajo el STOP o a una distancia inferior a un +5%

c3 = Criterio < 5

SCREENER [c1 AND c2 AND c3](Criterio)

6. Triple pantalla bajista

Este ProScreener se compone de tres condiciones en varias unidades de tiempo:

- **Condición 1:** MACD en vista semanal > 0
- **Condición 2:** MACD **decreciente** en vista semanal
- **Condición 3:** Estocástico en vista diaria > 70

Buscamos los valores más cercanos del nivel e compra STOP recomendado por el sistema (en este caso, se tratará del valor mínimo de la víspera).

El ProScreener mostrará los valores situados por encima de este nivel, así como aquellos que acaben de cruzarlo y que se mantienen en un margen cercano (no superior a un -5%).

Los resultados respetarán las condiciones en las unidades de tiempo indicadas (vista semanal y diaria).

Ejemplo: Triple pantalla bajista

REM Condición 1: MACD semanal > 0 y decreciente

TIMEFRAME(weekly)

MiMACD = MACD [12,26,9](Close)

c1 = MiMACD > 0 AND MiMACD < MiMACD [1]

REM Condición 2: Estocástico diario > 70

TIMEFRAME(daily)

MiSTO = Stochastic [14,3](Close)

c2 = MiSTO > 70

REM Nivel del stop de venta

MiStop = Low[1]

REM Criterio: Posición de los precios respecto al STOP

Criterio = (Close / MiStop - 1) * 100

REM Condición 3: los precios bajo el STOP o a una distancia inferior a un +5%

c3 = Criterio > -5

SCREENER [c1 AND c2 AND c3](Criterio)

< **Contacta con nosotros** >

> **No clientes**

☎ **914 890 888**

✉ informacion@selfbank.es

> **Clientes**

☎ **914 890 888**

✉ Área Clientes (Centro de comunicaciones > opción "Mensajes")

SelfBank
by Singular Bank